DEPARTMENT OF
~~NATIONAL RESOURCES~~
NATIONAL DEVELOPMENT

BUREAU OF MINERAL RESOURCES, GEOLOGY AND GEOPHYSICS

Record 1979/29

GRAV3 : AN INTERACTIVE COMPUTER PROGRAM TO CALCULATE THE GRAVITY ANOMALY OF A FINITE HORIZONTAL PRISM

by

R.D. Ogilvy

Record 1979/29

GRAV3 : AN INTERACTIVE COMPUTER PROGRAM TO CALCULATE THE GRAVITY ANOMALY OF A FINITE HORIZONTAL PRISM

by

R.D. Ogilvy

CONTENTS

		Page
	SUMMARY	
1.	INTRODUCTION	1
2.	CO-ORDINATE SYSTEM AND FORMULAE	2
3.	PROGRAM DESCRIPTION	4
	Flow chart	4
	Data input	4
	Loading the program	5
	Running the program	5
	Data output	6
4.	PROGRAM TESTING	6
5.	EXAMPLES	7
6.	REFERENCES	7
	APPENDIX : Program listing	

ILLUSTRATIONS

Figure 1 Co-ordinate system and notations

Figure 2 Program flow chart

Figure 3 Teletype input/output

Figure 4 Program testing

Figure 5 Examples

SUMMARY

A description is given of a computer program (GRAV3) used for calculating the gravity anomaly due to a finite prism. The program is written in FORTRAN IV for interactive use with an HP 21MX mini-computer and GOULD 5000 printer/plotter under RTE II using TEROS.

The program may be compiled, loaded, and executed using a system teletype. Little or no data preparation is required and interpretation is achieved by forward interactive modelling.

The program provides a computer plot together with suitable annotation and a cross-section of the case modelled.

1. INTRODUCTION

In metalliferous exploration the need commonly arises to interpret gravity anomalies which are caused by tabular bodies of finite strike length. The horizontal prism of trapezium cross-section provides the interpreter with a simple yet realistic geometrical form for a first-order interpretation of an anomaly.

A computer program is described for calculating the gravity anomaly of a finite homogeneous horizontal prism. The program is called GRAV3 and is based on formulae derived by Hjelt (1974).

The program has been written in FORTRAN IV for interactive use with an HP 21MX mini-computer and a GOULD 5000 printer/plotter under RTE II using TEROS. The basic program algorithm is simple and rapid, and could be readily adapted for use with some of the more recent desk top calculators. Little or no data preparation is required, and interpretation is achieved by forward interactive modelling.

2. CO-ORDINATE SYSTEM AND FORMULAE

Co-ordinate system

A rectangular co-ordinate system is used (Fig.1) with the X axis perpendicular to the strike direction, the Y axis parallel to the strike, and the Z axis positive downwards. The dipping surfaces of the prism are defined by the angles ξ and θ. A special case of the model is that of a prism in which $\xi = \theta$. The top and bottom faces of the structure are horizontal, and the two faces perpendicular to the strike direction are vertical.

The corners of the prism are defined by their X,Y,Z co-ordinates (e.g. X1, Y1, Z1) and the distances between the prism and the observation point P(X, Y, Z) are denoted by U, V, and W as shown in Figure 1.

Oblique profiles

Profiles, other than those normal to the prism strike, can be run by defining a profile co-ordinate system, at an angle β to the prism co-ordinate system (see Fig. 1b). In such cases, program GRAV3 plots the calculated gravity values with respect to the profile co-ordinate system.



Plan

Cross-section

a) Horizontal prism



Plan

b) Oblique profile

Fig. 1 Co-ordinate system and notations

Mathematical formulae

The gravity anomaly of a finite dipping prism ($\theta = \xi$) is given by Hjelt (1974):

$$\Delta g = G.\Delta\rho.To$$

where $\Delta\rho$ is the density contrast, G the gravitational constant, and

$$To = \left[\left[\left[(\Phi_1 + \Phi_2 + \Phi_3 + \Phi_4 + \Phi_5)\right]_{U_2}^{U_1}\right]_{V_2}^{V_1}\right]_{W_2}^{W_1}$$

where

$$\Phi_1 = W \arctan \frac{(UV)}{WR}$$

$$\Phi_2 = -p.\ \sin \xi \cos \xi \arctan \frac{(V.q)}{p.R}$$

$$\Phi_3 = -p.\ \sin^2 \xi \ \ln (V + R)$$

$$\Phi_4 = -V.\ \ln (U + R)$$

$$\Phi_5 = V.\ \cos \xi \ \ln (q.\ \sin \xi + R)$$

$$R = \sqrt{U^2 + V^2 + W^2}$$

$$p = U - W \cot \xi$$

$$q = W + U \cot \xi$$

The gravity anomaly of a prism with a more general trapezium cross-section is obtained by substituting θ for ξ when $U = U(2)$ or $u(2)^1$ (see Fig.1).

3. PROGRAM DESCRIPTION

Flow chart

The essential elements of the program structure are shown in the flow chart (Fig. 2). Program variables used in the flow chart are defined in the program listing (see Appendix). For operational convenience in BMR, a number of system subroutines are used, but these do not detract from the generality of the basic program algorithm. System subroutines are referred to by name (e.g. RMPAR, NUMB, etc.) and further details may be obtained from the HP-RTE II system manual and the GOULD printer manual.

Data input

The input model parameters used in the program are defined as follows:

MN	=	Model number. Always start with MN = 1
X1	=	X co-ordinate of top edge nearest origin
X2	=	X co-ordinate of top edge farthest from origin
Y1	=	Y co-ordinate of top edge nearest origin.
Y2	=	Y co-ordinate of top edge farthest from origin
Z1	=	Depth to top of prism
Z2	=	Depth to base of prism
XN	=	X co-ordinate of first observation point
YN	=	Y " " " " "
ZN	=	Z " " " " "
DX	=	Station increment along X axis
THETA	=	Dip of prism edge nearest to origin in degrees
ETA	=	" " " " farthest from origin in degrees
BETA	=	Angle between X axis and profile, measured clockwise from the positive X axis in degrees
DEN	=	Density contrast
NXN	=	Number of stations

ᵹ



Fig. 2 Program flow chart

The program is structured for maximum interactive use by the interpreter, and may be compiled, loaded, and executed by means of a system teletype. Model parameters for each case are fed to the computer using a free-field format.

The source program is stored as a resident disc file GRAV3S but is also available on paper tape.

Loading the program

To load the program, the following command sequence is observed:

1.	*GF, GRAV3S	This command calls the source program into the user scratch region.
2.	*FT, FL	This command causes the program to be compiled as a binary relocatable file.
3.	*GF, STAT..	This command schedules the binary file for loading.
4.	*LD, 0	This command causes the program to be loaded.

The above steps are once-only commands and need not be repeated while the system is up.

Running the program

To run the program, the following command sequence is observed:

1.	*LG, GRAV3	This command causes scheduling of the program for execution.
2.	"MN, X1, X2, Y1, Y2, Z1, Z2 = "	At this request, type in the appropriate model parameters.
3.	"XN, YN, ZN, DX, NXN, DEN, ETA, THETA, BETA = "	Type in appropriate parameters separated by a comma.
4.	"MISTAKE IN INPUT DATA?"	Answer YES or NO. If YES, the program will return to step 2.

5. "ANY MORE CASES?" Answer YES or NO. If YES, the program will return to step 2 so that additional cases can be modelled.

The user may log into the program any number of times using the command *LG, GRAV3. User commands are preceded by a * as shown; terminal replies are given in inverted commas.

An example illustrating the use of the input commands is given in Figure 3.

If more than one case is to be modelled, only the parameters to be varied need be changed. For all parameters preceding those to be changed, type in commas. Following parameters may be ignored.

Data output

Two modes of data output are utilised. The calculated gravity values are sent to the user teletype for immediate perusal (Fig. 3) and also to the line printer for plotting. The graphical plot includes suitable annotation and a cross-section of the case modelled (see Figs. 4 and 5). A cross-section of the model is not produced for oblique profiles.

4. PROGRAM TESTING

The program was tested against Nagy's (1966) formula for the gravity effect of a right rectangular prism and with one of Hjelt's (1974) cases. The results shown in Figure 4 indicate an excellent fit between the GRAV3 results and the test data.

```
◆GF,GRAV3S
GRAV3S COPIED TO SCRATCH TRACKS,  LENGTH; 201 RECORDS, 2800 WORDS
#◆FT,FL
TEROS TERMINAL ASSEMBLY/COMPILATION SYSTEM. BINARY OUTFILE =  STAT2B
#◆GF,STAT2B
STAT2B COPIED TO SCRATCH TRACKS,  LENGTH; 90 RECORDS, 2800 WORDS
#◆LD,0
 /LOADR:LOAD
 /LOADR:GRAV3 READY
 /LOADR:$END
```

Loading program

```
#◆LG,GRAV3
GRAV3 ON

  MN,X1,X2,Y1,Y2,Z1,Z2=
1,80.,100.,0.,100.,5.,60.

  XN,YN,ZN,DX,NXN,DEN,ETA,THETA,BETA=
0.,50.,0.,10.,21,2.5,45.,110.,0.

  MISTAKE IN INPUT DATA?
NO

1       MODEL NO.          1
0       STATION            GRAVITY VALUE
0            .00              .16
0          10.00              .20
0          20.00              .27
0          30.00              .36
0          40.00              .49
0          50.00              .67
0          60.00              .94
0          70.00             1.33
0          80.00             1.89
0          90.00             2.25
0         100.00             2.18
0         110.00             1.80
0         120.00             1.43
0         130.00             1.12
0         140.00              .86
0         150.00              .66
0         160.00              .50
0         170.00              .38
0         180.00              .29
0         190.00              .22
0         200.00              .17

  ANY MORE CASES?
YES
              ETC.ETC.#
```

Running program

Fig.3 Teletype input/output



Fig. 4 Program testing



Fig.5 Examples

5. EXAMPLES

Some further examples of the use of the program GRAV3 are shown in Figure 5.

6. REFERENCES

HJELT, S.-E., 1974 - The gravity anomaly of a dipping prism. Geoexploration, v.12(1), pp.29-39.

NAGY, D., 1966 - The gravitational attraction of a right rectangular prism. Geophysics, v.31, pp.362-371.

Appendix: PROGRAM LISTING

```
GRAV3
oTERMINAL      31
OUTPUT LOGICAL UNIT =       6
```

```
FTN4,B,L
      PROGRAM GRAV3
C      AUTHOR- R.D.OGILVY
C      THIS PROGRAM CALCULATES THE GRAVITY ANOMALY OF A
C       3-DIMENSIONAL DIPPING PRISM, OR  TRAPEZOIDAL STRUCTURE.
C      ALL DIMENSIONS ARE IN METRES.
C      MN=MODEL NUMBER
C      X1=CO-OD OF FIRST TOP EDGE ON X AXIS
C      X2=CO-OD OF SECOND TOP EDGE ON X AXIS
C      Y1=CO-OD OF FIRST EDGE ON Y AXIS
C      Y2=CO-OD OF SECOND EDGE ON Y AXIS
C      Z1=DEPTH TO TOP OF PRISM
C      Z2=DEPTH TO BASE OF PRISM
C      XN=X CO-OD OF FIRST FIELD POINT
C      YN=Y CO-OD OF FIRST FIELD POINT
C      ZN=Z CO-OD OF FIRST FIELD POINT
C
C      ALL DISTANCES MEASURED IN METRES.
C      ALL ANGLES MEASURED IN DEGREES.
C      DX=STATION INCREMENT ALONG X AXIS
C      THETA =DIP OF PRISM(OR TRAPEZOID ) EDGE NEAREST TO ORIGIN.
C      ETA=DIP OF PRISM EDGE FURTHEST FROM ORIGIN .
C      BETA=ANGLE BETWEEN X AXIS AND PROFILE, MEASURED CLOCKWISE
C      FROM +VE X AXIS.
C      DEN=DENSITY CONSTRAST( C.G.S)
C      GC=GRAVITATIONAL CONSTANT (C.G.S)
C      NXN=NUMBER OF STATIONS
C      AG=ARRAY FOR GRAVITY VALUES
C      STAT=ARRAY FOR OBSERVATION POINTS
C
      DIMENSION AG(400),STAT(400),F1(8),F2(8),F3(8),F4(8),LU(5)
      DIMENSION F5(8),U(2),V(2),W(2)
      INTEGER MN,NXN
      REAL X1,X2,Y1,Y2,Z1,Z2,XN,YN,ZN,DX,DEN,ETA,THETA,BETA
      INTEGER STR1(10),STR2(7),STR3(8),STR4(8),STR5(11),
     *STR6(8),STR7(8),STR8(8),STR10(7),STR11(7),STR9(6)
      DATA STR1/2HGR,2HAV,2HIT,2HY ,2HVA,2HLU,2HE(,2HMG,2HAL,2HS)/
      DATA STR2/2HDE,2HPT,2HH(,2HME,2HTR,2HES,2H) /
      DATA STR3/2HDI,2HST,2HAN,2HCE,2H(M,2HET,2HRE,2HS)/
      DATA STR4/2HMO,2HDE,2HL ,2HNO,2H. ,2H  ,2H  ,2H  /
      DATA STR5/2HDE,2HNS,2HIT,2HY ,2HCO,2HNT,2HRA,2HST,2H  ,2H  ,2H  /
      DATA STR6/2HMO,2HDE,2HL ,2HDE,2HPT,2HH ,2H  ,2H  /
       DATA STR7/2HDI,2HP ,2HAN,2HGL,2HE(,2HET,2HA),2H  /
      DATA STR8/2HDI,2HP ,2HAN,2HGL,2HE(,2HTH,2HET,2HA)/
      DATA STR9/2HPR,2HOF,2HIL,2HE ,2H(Y,2HN)/
      DATA STR10/2HPR,2HOF,2HIL,2HE ,2HAN,2HGL,2HE /
      DATA STR11/2HCR,2HOS,2HS-,2HSE,2HCT,2HIO,2HN /

C      READ INPUT DATA
C
      CALL RMPAR(LU)
    1 WRITE(LU,300)
  300 FORMAT(/2X,"MN,X1,X2,Y1,Y2,Z1,Z2=")
      READ(LU,*)MN,X1,X2,Y1,Y2,Z1,Z2
```

```
      WRITE(LU,301)
  301 FORMAT(/2X,"XN,YN,ZN,DX,NXN,DEN,ETA,THETA,BETA=")
      READ(LU,*)XN,YN,ZN,DX,NXN,DEN,ETA,THETA,BETA
   14 WRITE(LU,303)
  303 FORMAT(/2X,"MISTAKE IN INPUT DATA?")
      READ(LU,6)NGOOD
    6 FORMAT(A2)
      IF(NGOOD.EQ.2HYE)GO TO 1
C
C      CONVERT INPUT DATA
C
      GC=6.67/100000000
      PI=3.14159
      ETA=ETA*PI/180
      THETA=THETA*PI/180
      BETA=BETA*PI/180
      XG=XN
C
      DO 99 L=1,NXN
      X=(L-1)*DX*COS(BETA)+XN
      Y=(L-1)*DX*SIN(BETA)+YN
       STAT(L)=XN+(L-1)*DX
      H=Z2-Z1
      U(1)=X-X2
      U(2)=X-X1
      V(1)=Y-Y2
      V(2)=Y-Y1
      W(1)=ZN-Z2
      W(2)=ZN-Z1
C
C      CALCULATE GRAVITY FUNCTION G
C
      M=1
      G=0
      DO 10 I=1,2
      DO 10 J=1,2
      DO 10 K=1,2
      IF((ABS(W(K))-0.001).LT.0)W(K)=0.001
      IF(I.EQ.1)PHI=ETA
      IF(I.EQ.2)PHI=THETA
      IF(K.EQ.1)U(I)=U(I)-H/TAN(PHI)
      R=SQRT(U(I)*U(I)+V(J)*V(J)+W(K)*W(K))
      P=U(I)-W(K)/TAN(PHI)
      IF((ABS(P)-0.001).LT.0)P=0.001
      Q=W(K)+U(I)/TAN(PHI)
      F1(M)=W(K)*ATAN((V(J)*U(I))/(W(K)*R))
      F2(M)=-P*SIN(PHI)*COS(PHI)*ATAN((V(J)*Q)/(P*R))
      F3(M)=-P*SIN(PHI)*SIN(PHI)*ALOG(V(J)+R)
      F4(M)=-V(J)*ALOG(U(I)+R)
      F5(M)=V(J)*COS(PHI)*ALOG(Q*SIN(PHI)+R)
      IJK=I+J+K
      IF(IJK.EQ.4.OR.IJK.EQ.6)GO TO 19
    5 F1(M)=-F1(M)
      F2(M)=-F2(M)
      F3(M)=-F3(M)
```

```
      F4(M)=-F4(M)
      F5(M)=-F5(M)
   19 IF(K.EQ.1)U(I)=U(I)+H/TAN(PHI)
      M=M+1
   10 CONTINUE
      DO 11 N=1,8
      G=G+F1(N)+F2(N)+F3(N)+F4(N)+F5(N)
   11 CONTINUE
C
C     CALCULATION OF GRAVITY VALUE
C
      AG(L)=-100000.0*GC*DEN*G
   99 CONTINUE
      XN=XG
C
C      PRINT RESULTS
C
      WRITE(LU,3)MN
C     WRITE(6,3)MN
    3 FORMAT(1H1,7X,9HMODEL NO.,I10)
      WRITE(LU,202)
C     WRITE(6,202)
  202 FORMAT(1H0,7X,7HSTATION,10X,13HGRAVITY VALUE)
      DO 56 J=1,NXN
      WRITE(LU,60)STAT(J),AG(J)
C     WRITE(6,60)STAT(J),AG(J)
   60 FORMAT(1H0,5X,F10.2,5X,F10.2)
   56 CONTINUE
C
C     PLOT AXIS
C
      DIST=DX*FLOAT(NXN)/10.
      SI2=-Z2/3.
      CALL PLOTS(17.,10.,0)
      CALL FACTR(0.5)
      IF(MN.EQ.1)CALL PLOT(1.,0.5,-3)
      CALL AXIS(0.,4.,STR1,20,5.,90.,0.,0.5)
      CALL AXIS(0.,0.,STR2,14,3.,90.,Z2,SI2)
      CALL AXIS(0.,3.,STR3,16,10.,0.,XG,DIST)
C
C      PLOT POINTS
C
      CALL PLOT(0.,4.,-3)
      XP=0.
      DO 600 I=1,NXN
      YP=AG(I)/SCM
      CALL SYMB(XP,YP,0.1,7,0.,-1,0.1)
      XP=XP+DX/DIST
  600 CONTINUE
      BETA=BETA*180./PI
      IF(IFIX(BETA).GT.0)GO TO 900
      XP1=X1/DIST-XG/DIST
      YP1=Z1/SI2
      XP2=X2/DIST-XG/DIST
      XP3=(X2+H/TAN(ETA))/DIST-XG/DIST
```

```
      XP4=(X1+H/TAN(THETA))/DIST-XG/DIST
      CALL PLOT(0.,-1.,-3)
      CALL PLOT(XP1,YP1,3)
      CALL PLOT(XP2,YP1,2)
      CALL PLOT(XP3,-3.,2)
      CALL PLOT(XP4,-3.,2)
      CALL PLOT(XP1,YP1,2)
      CALL SYMB(7.,-1.5,0.1,STR11,0.,13)
  900 THETA=THETA*180./PI
      ETA=ETA*180./PI
      CALL SYMB(7.,5.5,0.1,STR4,0.,16)
      CALL SYMB(7.,5.,0.1,STR5,0.,22)
      CALL SYMB(7.,4.7,0.1,STR6,0.,16)
      CALL SYMB(7.,4.4,0.1,STR7,0.,16)
      CALL SYMB(7.,4.1,0.1,STR8,0.1,16)
      CALL SYMB(7.,3.8,0.1,STR9,0.,12)
      CALL SYMB(7.,3.5,0.1,STR10,0.,14)
      CALL NUMB(8.8,5.5,0.1,FLOAT(MN),0.,0)
      CALL NUMB(8.8,5.0,0.1,DEN,0.,4)
      CALL NUMB(8.8,4.7,0.1,Z1,0.,2)
      CALL NUMB(8.8,4.4,0.1,ETA,0.,2)
      CALL NUMB(8.8,4.1,0.1,THETA,0.,2)
      CALL NUMB(8.8,3.8,0.1,YN,0.,2)
      CALL NUMB(8.8,3.5,0.1,BETA,0.,2)
C
C       TERMINATE PLOT
C
   27 CALL PLOT(0.,0.,999)
      CALL GOPLT
   75 WRITE(LU,20)
   20 FORMAT(/2X,"ANY MORE CASES?")
      READ(LU,30)MORE
   30 FORMAT(A2)
      IF(MORE.EQ.2HYE)1,27
      STOP
      END
      END$
```